```
p = {(3 + Cos[u]) * Cos[v], (3 + Cos[u]) * Sin[v], Sin[u]}
```

```
{(3 + Cos[u]) Cos[v], (3 + Cos[u]) Sin[v], Sin[u]}
```

```
ParametricPlot3D[p, {u, -Pi, Pi}, {v, -Pi, Pi},
 Boxed → False, Axes → False, BoxRatios → Automatic, Mesh → {5, 5},
 PlotStyle → {Pink, Specularity[White, 20], Opacity[.8]}, Lighting → "Neutral",
 PerformanceGoal → "Quality", PlotRange → {{-5, 5}, {-5, 5}, {-5, 5}}]
```



```
(* parcialni derivace *)
```

```
pu = D[p, u]
pv = D[p, v]
```

```
{-Cos[v] Sin[u], -Sin[u] Sin[v], Cos[u]}
```

```
{-(3 + Cos[u]) Sin[v], (3 + Cos[u]) Cos[v], 0}
```

```
(* I. forma plochy *)
```

```
G = {{Simplify[pu.pu], Simplify[pu.pv]}, {Simplify[pv.pu], Simplify[pv.pv]}}
MatrixForm[%]
```

```
{{1, 0}, {0, (3 + Cos[u])^2}}
```

$$\begin{pmatrix} 1 & 0 \\ 0 & (3+\text{Cos}[u])^2 \end{pmatrix}$$

```
(* křivka v mapě *)
```

```
ccc = {1 + t, 2 + 2 * t^2}
```

$$\{1+t,\ 2+2\,t^2\}$$

```
(* tatáž křivka na ploše *)
c = p /. {u → ccc[[1]], v → ccc[[2]]}
{(3 + Cos[1 + t]) Cos[2 + 2 t^2], (3 + Cos[1 + t]) Sin[2 + 2 t^2], Sin[1 + t]}
Show[ParametricPlot3D[p, {u, -Pi, Pi}, {v, -Pi, Pi},
  Boxed → False, Axes → False, BoxRatios → Automatic, Mesh → {5, 5},
  PlotStyle → {Pink, Specularity[White, 20], Opacity[.8]}, Lighting → "Neutral",
  PerformanceGoal → "Quality", PlotRange → {{-5, 5}, {-5, 5}, {-5, 5}}],
 ParametricPlot3D[c, {t, -0.5, 0.5}]]
```



```
(* tečný vektor *)
dccc = D[ccc, t]
{1, 4 t}
(* I. forma podél křivky *)
Gt = G /. {u → ccc[[1]], v → ccc[[2]]}
{{1, 0}, {0, (3 + Cos[1 + t])^2}}
{{1, 0}, {0, (3 + Cos[1 + t])^2}}
MatrixForm[%]
{{1, 0}, {0, (3 + Cos[1 + t])^2}}
(1          0
 0  (3 + Cos[1 + t])^2)
```

```
(* délka křivky na ploše *)
NIntegrate[Sqrt[dccc.Gt.dccc], {t, -0.5, 0.5}]
```

3.73072

```
(* délka křivky v mapě *)
NIntegrate[Sqrt[dccc.dccc], {t, -0.5, 0.5}]
```

1.47894

```
(* plocha toru *)
Integrate[Sqrt[Det[G]], {u, 0, 2 Pi}, {v, 0, 2 Pi}]
```

$12\,\pi^2$

```
NN = Simplify[Normalize[Cross[pu, pv]], Assumptions → {u > 0, v > 0}]
```

```
{-Cos[u] Cos[v], -Cos[u] Sin[v], -Sin[u]}
```

```
puu = D[pu, u]
puv = D[pu, v]
pvv = D[pv, v]
```

```
{-Cos[u] Cos[v], -Cos[u] Sin[v], -Sin[u]}
```

```
{Sin[u] Sin[v], -Cos[v] Sin[u], 0}
```

```
{-(3 + Cos[u]) Cos[v], -(3 + Cos[u]) Sin[v], 0}
```

```
(* II. forma plochy *)
H = Simplify[{{puu.NN, puv.NN}, {puv.NN, pvv.NN}}]
```

```
{{1, 0}, {0, Cos[u] (3 + Cos[u])}}
```

```
MatrixForm[%]
```

$$\begin{pmatrix} 1 & 0 \\ 0 & \text{Cos}[u]\ (3+\text{Cos}[u]) \end{pmatrix}$$

```
K = Det[H] / Det[G]
```

$$\frac{3\,\text{Cos}[u] + \text{Cos}[u]^2}{9 + 6\,\text{Cos}[u] + \text{Cos}[u]^2}$$

```
ParametricPlot3D[p, {u, -Pi, Pi}, {v, -Pi, Pi}, Boxed → False, Axes → False,
 BoxRatios → Automatic, Mesh → {5, 5}, PlotRange → {{-5, 5}, {-5, 5}, {-5, 5}},
 ColorFunction → Function[{x, y, z, u, v}, Hue[K]], ColorFunctionScaling → False]
```



```
W = Inverse[G].H
```

$$\left\{\left\{\frac{(3+\text{Cos}[u])^2}{9+6\,\text{Cos}[u]+\text{Cos}[u]^2},\ 0\right\},\ \left\{0,\ \frac{\text{Cos}[u]\ (3+\text{Cos}[u])}{9+6\,\text{Cos}[u]+\text{Cos}[u]^2}\right\}\right\}$$

```
(* hlavní směry a hlavní křivosti *)

Eigensystem[W]
```

$$\left\{\left\{\frac{\text{Cos}[u]}{3+\text{Cos}[u]},\ 1\right\},\ \{\{0,\ 1\},\ \{1,\ 0\}\}\right\}$$